

Polarkoordinaten Einführung: Archimedische Spirale

Prof. Dr. Dörte Haftendorn: Mathematik mit MuPAD 4, Mrz. 06 (Version 3 ex.) Update 10.01.07

Web: www.mathematik-verstehen.de

<http://haftendorn.uni-lueneburg.de>

Level 1

Archimedische Spirale:

t ist der Polarwinkel, ihm ist der Radius proportional:

```
r:=t->t/4:r(t);
```

$$\frac{t}{4}$$


Der Punkt mit den "Polarkoordinaten" r und t ist : P=(r,t), wobei r von t abhängt.

P=(t,r) wäre dabei eigentlich logischer, aber das Obige ist das Übliche.

Gibt man für t einen Bereich an, kann man ein Graphik-Element erzeugen, archi genannt. plot(archi) zeichnet dann das Graphik-Element.

```
archi:=plot::Polar([r(t),t],t=0..20,LineWidth=1,Mesh=400);
plot(archi);
```

```
plot::Polar([t/4,t],t=0..20)
```




Eindrucksvoller ist, wenn die Spirale langsam entsteht.


Eine solche Animation ist in MuPAD 3 ganz einfach.

Zusätzlich ist noch der Polarradius eingezeichnet.

```
archi:=plot::Polar([r(t),t],t=0..ende,ende=0..20,
    LineWidth=1,LineColor=RGB::Red,Mesh=400):
radius:=plot::Line2d([0,0],[r(t)*cos(t),r(t)*sin(t)],t=0..20):
plot(archi,radius,AnimationStyle=BackAndForth)
```

1





Damit man die Animation sieht, muss man doppelt in die Zeichnung klicken und dann evt. den oben erscheinenden Player bedienen, angezeigt durch:



animieren durch Anklicken!
Durch Eintragen anderer Funktionsterme für $r(t)$ ganz oben, kann man schon Vieles erkunden.

LEVEL 2

Nun wird dazu noch die gekoppelte kartesische Darstellung gezeichnet. Sie dazu die ausführliche Seite : Polar-Kartesisch.

```

archi:=plot::Polar([r(t),t],t=0..ende,ende=0..20,
    LineWidth=1,LineColor=RGB::Red, Mesh=400):
archikart:=plot::Curve2d([t,r(t)],t=0..ende,ende=0..20,
    LineWidth=1, Mesh=400, LineColor=RGB::Green):
radius:=plot::Line2d([0,0],[r(t)*cos(t),r(t)*sin(t)],t=0..20):
radiusordi:=plot::Line2d([t,0],[t,r(t)],t=0..20):
plot(archi,radius,archikart,radiusordi, Scaling=Constrained,
    AnimationStyle=BackAndForth);

```

